Egyptian multiplication

Multiplication by doubling, halving and some bookkeeping.
Recursive and iterative versions

1 Recursive version

```
(%i1) Egyptian_multiplication(a, b) :=
        (if verbose then printf(true, "~10d ~10d  ~%",  a, b),
         if a=1
           then b
           else (if mod(a, 2) = 0
                   then Egyptian_multiplication(a/2, 2*b)
                   else b + Egyptian_multiplication(a-1, b) ) ) $ ;

(%i2) verbose : true    /* global control variable */  $

(%i3) Egyptian_multiplication(60,4);
        60        4
        30        8
        15        16
        14        16
        7         32
        6         32
        3         64
        2         64
        1         128
(%o3) 240

(%i4) verbose : false $

(%i5) Egyptian_multiplication(80,4);
(%o5) 320

(%i6) (verbose : true, Egyptian_multiplication(92, 15) ) ;
        92        15
        46        30
        23        60
        22        60
        11        120
        10        120
        5         240
        4         240
        2         480
        1         960
(%o6) 1380
```

```
(%i7) (verbose : false, Egyptian_multiplication(92, 15) ) ;
(%o7) 1380
```

2 Iterative version

```
(%i8) Egyptian_multiplication_iterative(a,b) :=
        block([a1 : a, b1 : b, c : 0 ],
          while a1 > 1 do
           (if verbose then printf(true, "~10d ~10d ~10d ~%", a1, b1, c),
            if mod(a1, 2)=0
              then (a1 : a1/2, b1 : b1*2)
              else (a1 : a1-1, c : c+b1) ),
          if verbose then printf(true, "~10d ~10d ~10d ~%", a1, b1, c),
          b1+c ) $ ;

(%i9) (verbose : true, Egyptian_multiplication_iterative(92,15)) ;
        92        15         0
        46        30         0
        23        60         0
        22        60        60
        11       120        60
        10       120       180
         5       240       180
         4       240       420
         2       480       420
         1       960       420
(%o9) 1380

(%i10) (verbose : false, Egyptian_multiplication_iterative(92,5)) ;
(%o10) 460
```